

# EvoSim Grammar

## Lexer

### literals

+ - \* / % ( ) ^ @ < > { } = , . [ ] '

### reserved

OR AND NOT TRUE FALSE WORLD WIDTH HEIGHT INFINITE DEFAULT AT SIZE TERRAIN  
SIMULATION EPISODES STOP MAX\_ROUNDS IF ELSE GENE HEALTH HUNGER LEGS EYES  
ARMS HORNS SMELL FINS NOSE MOUTH SMELLING VISION MOVE EAT REPRODUCE ATTACK  
DEFEND PICK SWIM VALUE IN MUTATION CHANCE STEP LOOP CONTINUE BREAK  
ACTIONS\_TIME AVAILABLE\_COMMANDS FUNC RETURN DNA COST ENTITY COEXISTENCE REPR  
ORGANISM BEHAVIOR DECIDE TIME

### tokens

INTDIV EQ NEQ GE LE NUMBER STRING ID

## Parser

program  $\Rightarrow$  gene\_stmt\_list dna\_stmt\_list behavior\_stmt\_list entity\_org\_stmt\_list  
world\_stmt newline sim\_stmt maybe\_nl

gene\_stmt\_list  $\Rightarrow$  gene\_stmt gene\_stmt\_list | epsilon

gene\_stmt  $\Rightarrow$  GENE phygene\_stmt | GENE percpgene | GENE actgene\_stmt

percpgene  $\Rightarrow$  SMELLING | VISION

actgene\_stmt  $\Rightarrow$  actgene ID { COST NUMBER }

actgene  $\Rightarrow$  MOVE | EAT | REPRODUCE | ATTACK | DEFEND | PICK | SWIM

phygene\_stmt  $\Rightarrow$  phygene ID { phygeneprop phygeneprop }

phygene  $\Rightarrow$  HEALTH | HUNGER | LEGS | EYES | ARMS | HORNS | SMELL | FINS | NOSE | MOUTH

phygeneprop  $\Rightarrow$  VALUE NUMBER IN { NUMBER NUMBER } | MUTATION { mutationprop mutationprop }

mutationprop  $\Rightarrow$  CHANCE NUMBER | STEP NUMBER

dna\_stmt\_list  $\Rightarrow$  dna\_stmt dna\_stmt\_list | epsilon

dna\_stmt  $\Rightarrow$  DNA ID { dna\_elem\_list }

dna\_elem\_list  $\Rightarrow$  dna\_elem dna\_elem\_list | epsilon

dna\_elem  $\Rightarrow$  SMELLING | VISION | ID | DNA ID

behavior\_stmt\_list  $\Rightarrow$  behavior\_stmt behavior\_stmt\_list | epsilon

behavior\_stmt  $\Rightarrow$  BEHAVIOR ID { func\_stmt\_list decide\_stmt } | BEHAVIOR ID

decide\_stmt  $\Rightarrow$  DECIDE ORGANISM TIME { stmt\_list }

entity\_org\_stmt\_list  $\Rightarrow$  entity\_stmt entity\_org\_stmt\_list | organism\_stmt entity\_org\_stmt\_list | epsilon

entity\_stmt  $\Rightarrow$  ENTITY { entityprop entityprop entityprop }

entityprop  $\Rightarrow$  COEXISTENCE bool | REPR ID | AT { position position\_list }

organism\_stmt  $\Rightarrow$  ORGANISM { orgprop orgprop orgprop orgprop }

orgprop  $\Rightarrow$  REPR ID | DNA ID | BEHAVIOR ID | AT { position position\_list }

position\_list  $\Rightarrow$  position position\_list | epsilon

position  $\Rightarrow$  ( NUMBER NUMBER )

world\_stmt  $\Rightarrow$  WORLD { worldprop worldprop }

worldprop  $\Rightarrow$  SIZE worldsize | TERRAIN worldterrain

worldsize  $\Rightarrow$  INFINITE { worldsizeprop worldsizeprop } | { worldsizeprop worldsizeprop }

worldsizeprop  $\Rightarrow$  HEIGHT NUMBER | WIDTH NUMBER

worldterrain  $\Rightarrow$  { terrainprop\_list }

terrainprop\_list  $\Rightarrow$  terrainprop terrainprop\_list | epsilon

terrainprop  $\Rightarrow$  DEFAULT ID | ID AT { NUMBER terrainposn\_list } | ID

terrainposn\_list  $\Rightarrow$  NUMBER terrainposn\_list | epsilon

```

sim_stmt ⇒ SIMULATION { simprop simprop simprop simprop simprop }

simprop ⇒ EPISODES NUMBER | MAX_ROUNDS NUMBER | STOP SIMULATION { stmt_list } |
ACTIONS_TIME NUMBER | AVAILABLE_COMMANDS { command_list }

command_list ⇒ ID command_list | epsilon

func_stmt_list ⇒ func_stmt func_stmt_list | epsilon

func_stmt ⇒ FUNC ID = param_list { stmt_list }

param_list ⇒ ID param_list | epsilon

ret_stmt ⇒ RETURN disjunction | RETURN epsilon

stmt_list ⇒ stmt stmt_list | epsilon

stmt ⇒ if_stmt | var_stmt ; | index_stmt ; | loop_stmt ; | CONTINUE ; | BREAK ; |
ret_stmt ; | disjunction ;

var_stmt ⇒ accessing = disjunction

index_stmt ⇒ naming [ disjunction ] = disjunction

loop_stmt ⇒ LOOP loop_init , loop_condition , loop_set { stmt_list }

loop_condition ⇒ disjunction | epsilon

loop_init ⇒ var_stmt | epsilon

loop_set ⇒ var_stmt | epsilon

if_stmt ⇒ IF disjunction { stmt_list } else_stmt

else_stmt ⇒ ELSE if_stmt | ELSE { stmt_list } | epsilon

disjunction ⇒ conjunction OR conjunction | conjunction

conjunction ⇒ negation AND negation | negation

negation ⇒ NOT comparison | comparison

comparison ⇒ expr EQ expr | expr NEQ expr | expr GE expr | expr LE expr | expr
< expr | expr > expr | expr

expr ⇒ expr + term | expr - term | term

term ⇒ term * factor | term / factor | term % factor | term INTDIV factor |
factor

```

factor  $\Rightarrow$  - factor | power

power  $\Rightarrow$  atom ^ factor | atom @ factor | naming

naming  $\Rightarrow$  naming [ disjunction ] | accessing | KEYS ( naming ) | ID ( arg\_list )  
| atom

accessing  $\Rightarrow$  word | word . accessing

word  $\Rightarrow$  SIMULATION | ORGANISM | ID

atom  $\Rightarrow$  NUMBER | STRING | bool | ( expr ) | [ arg\_list ] | { keyarg\_list }

bool  $\Rightarrow$  TRUE | FALSE

arg\_list  $\Rightarrow$  disjunction | disjunction , arg\_list | epsilon

keyarg\_list  $\Rightarrow$  keyarg | keyarg , keyarg\_list | epsilon

keyarg  $\Rightarrow$  disjunction = disjunction